

LES TOURS DE HANOÏ

par André Chéramy

Dans la série "Je fouille mes archives pour en faire profiter les copains", voici aujourd'hui un petit programme BASIC, mis au point il y a bien des années. Ce petit jeu n'a d'autre prétention que de vous distraire et de montrer qu'avec quelques lignes de BASIC on peut déjà bien s'amuser.

C'est le très classique jeu des Tours de Hanoï. Au départ, vous avez à gauche, une "tour" constituée de cinq disques empilés par taille décroissante (TOUR No1). Le but du jeu est de déplacer cette tour à droite (TOUR No3). La règle est simple : on ne peut déplacer les disques que un à un et il est interdit de placer un disque quelconque sur un disque plus petit. Pour déplacer un disque, donnez le numéro de la tour de départ, puis le numéro de la tour d'arrivée. Au fur et à mesure des déplacements, le nombre de coups déjà joués est affiché. Il y a des records à battre !

```
REGLE DU JEU DES TOURS DE HANOÏ
-----
Au départ, vous avez à gauche une
tour de cinq disques empilés par
taille décroissante. Il faut mettre
cette tour à droite. Vous pouvez
déplacer ces disques un à un, en
indiquant le No de la tour de départ
puis celui de la tour d'arrivée.
Mais ATTENTION Vous ne devez jamais
placer un grand disque sur un plus
petit!

      XXXX
     XXXX
    XXXX
   XXXX
  XXXX
 XXXX
-----
TOUR No1  TOUR No2  TOUR No3
-->
```

```
      XXXX
     XXXX
    XXXX
   XXXX
  XXXX
 XXXX
-----
TOUR No1  TOUR No2  TOUR No3
NOMBRE DE COUPS DEJA JUES: 0

TOUR DE DEPART
```

```
      XXXX
     XXXX
    XXXX
   XXXX
  XXXX
 XXXX
-----
TOUR No1  TOUR No2  TOUR No3
NOMBRE DE COUPS DEJA JUES: 1

TOUR DE DEPART1 1
TOUR D'ARRIVEE3 3
C'EST FAUX

      XXXX
     XXXX
    XXXX
   XXXX
  XXXX
 XXXX
-----
TOUR No1  TOUR No2  TOUR No3
NOMBRE DE COUPS DEJA JUES: 2

TOUR DE DEPART
```

```
      XXXX
     XXXX
    XXXX
   XXXX
  XXXX
 XXXX
-----
      XX
-----
TOUR No1  TOUR No2  TOUR No3
NOMBRE DE COUPS DEJA JUES: 39

TOUR DE DEPART1 1
TOUR D'ARRIVEE3 3

      XXXX
     XXXX
    XXXX
   XXXX
  XXXX
 XXXX
-----
TOUR No1  TOUR No2  TOUR No3
NOMBRE DE COUPS DEJA JUES: 40

BRAVO !!!

Ready
```

```

10 ' ***** TOURS DE HANOI *****
11 ' V0302911820 André Chéramy
20 DIM T(5,3)
21 CLS:PAPER6:INK1:PRINT:PRINT:PRINT:PRINT
22 PRINT"Voulez vous la règle (O/N)?" :GET A$:PRINT A$
24 IF A$="O" OR A$="o" THEN HANOI.HLP:GET A$
25 CLS:PAPER6:INK1
30 FOR I=1 TO 5:T(I,1)=I:NEXT I
60 FOR J=2 TO 3:FOR I=1 TO 5:T(I,J)=0:NEXT I:NEXT J
110 C=0
120 GOSUB 260:GOSUB 500:IF BRAVO=1 THEN GOTO 410
125 PRINT "TOUR DE DEPART" :GET A$:PRINT A$ :D=ASC(A$)
126 IF (D<>49) AND (D<>50) AND (D<>51) AND (D<>47) THEN PRINT: GOTO 125
128 D=VAL(A$):PRINT D:IF D=0 THEN 410
130 PRINT "TOUR D'ARRIVEE" :GETA$:PRINT A$ :A=ASC(A$)
132 IF (A<>49) AND (A<>50) AND (A<>51) AND (A<>47) THEN PRINT: GOTO 130
134 A=VAL(A$):PRINT A:IF A=0 THEN 410
140 C=C+1
150 FOR I=1 TO 5
160 IF T(I,D)<>0 THEN Q=T(I,D):T(I,D)=0:GOTO 180
170 NEXT I
172 I=I-1
175 IF T(I,D)=0 THEN PRINT"C'EST FAUX": GOTO 120
180 FOR K=5 TO 1 STEP-1
190 IF T(K,A)=0 THEN 210
200 NEXT K
210 IF K=5 THEN T(K,A)=Q: GOTO 120
220 IF T(K+1,A)>Q THEN T(K,A)=Q : GOTO 120
230 PRINT"C'EST FAUX"
240 T(I,D)=Q
250 GOTO 120
255 CLS:PRINT:PRINT:PRINT:PRINT
260 REM AFFICHAGE TOURS
264 PRINT CHR$(17)
266 FOR I=1 TO 5
270 FOR J=1 TO 3
280 IF T(I,J)=0 THEN PRINT"          ";
290 IF T(I,J)=1 THEN PRINT"         XX ";
300 IF T(I,J)=2 THEN PRINT"        XXXX ";
310 IF T(I,J)=3 THEN PRINT"       XXXXXX ";
320 IF T(I,J)=4 THEN PRINT"      XXXXXXXX ";
330 IF T(I,J)=5 THEN PRINT"     XXXXXXXXXX ";
340 NEXT J
350 PRINT
360 NEXT I
370 PRINT "-----"
375 PRINT "  TOUR No1   TOUR No2   TOUR No3"
380 PRINT"          "
400 PRINT "NOMBRE DE COUPS DEJA JOUES:";C
402 PRINT"          "
404 PRINT CHR$(17)
405 RETURN
410 END
500 REM GAGNE?
505 BRAVO=0
510 FOR V=1 TO 5
520 IF T(V,3)<>V THEN RETURN
530 NEXT V
540 PRINT CHR$(140);CHR$(128);"BRAVO !!!"
550 PING:BRAVO=1
560 RETURN
600 !HANOI.HLP
650 GET A$
666 RETURN

```